

Análisis de transferencias de jugadores uruguayos a ligas europeas utilizando una base de datos de grafos

Autores: Christian Ferrari, Eduardo Gastelú
Supervisores: Lorena Etcheverry, Martin Giachino
Instituto de Computación
Facultad de Ingeniería, Universidad de la República
Montevideo, Uruguay

Resumen

El siguiente documento detalla el modelado en Neo4j de una base de datos de grafos, centrada en las transferencias de futbolistas Uruguayos en las ligas Premier League, Ligue 1 y La Liga entre 2010 y 2022. El objetivo es analizar los patrones de transferencia de estos jugadores de forma estructurada y detallada.

I. INTRODUCCIÓN

Este proyecto busca analizar las transferencias históricas de futbolistas vinculados a equipos Uruguayos que han competido y compiten en las Ligas Europeas. A través de una base de datos de grafos en Neo4j, se examina el desempeño de los jugadores, así como sus equipos, ciudades de origen y destino. Los datos fueron obtenidos de **Api-Sports**¹, los cuales comprenden las temporadas 2010-2022 de las Ligas Europeas Premier League, Ligue 1, La Liga, y las temporadas 2012-presente de las Ligas Uruguayas Primera División y Segunda División. Esta información incluye estadísticas y transferencias de los jugadores para cada temporada y equipo en el que hayan jugado.

II. OBTENCIÓN Y LIMPIEZA DE DATOS

El plan gratuito ofrecido por la API utilizada en este proyecto permitía un máximo de 100 consultas diarias por cuenta. A pesar de comenzar con tiempo y planificar cuidadosamente las consultas para la obtención de datos, fue necesario recurrir a múltiples cuentas, cambiar de IP y el uso de VPN para adquirir la cantidad mínima de datos requerida para construir el grafo.

Los datos obtenidos se almacenaron en **MongoDB**, aprovechando su eficiencia en el manejo de datos en formato **JSON**², formato en el cual la API retornaba los resultados. De esta forma, nuestro proyecto utilizó simultáneamente dos bases de datos. En **MongoDB**, se ejecutaron consultas para desglosar y modificar los datos, preparándolos para su posterior uso. Fueron eliminados campos que no eran de particular interés para nuestro estudio, así como aquellos con errores o nulos. Además de preparadas las colecciones para su posterior traspaso como nodos, facilitando su inserción y relación entre ellos. En resumen, se utilizó **MongoDB** para organizar y controlar la generación y modificación de los nodos y aristas de nuestro grafo.

El código fuente que se utilizó para automatizar las peticiones a la API, así como las colecciones y consultas realizadas se encuentran disponibles en el siguiente repositorio de GIT proporcionado por la **FING**, Proyecto final BDNR 2023 Grupo 21³

III. DISEÑO DEL MODELO DE GRAFOS

El diseño del modelo de grafos para este proyecto incorpora diversos nodos para representar entidades significativas en el análisis de las transferencias y el desempeño de los jugadores de fútbol.

El nodo **Player** contiene información acerca del jugador, incluyendo su `name` (nombre y apellido), `birth_date` (fecha de nacimiento), y `birth_place` (lugar de nacimiento). Cada jugador está vinculado a su país de origen a través de la relación **BORN_IN**, que conecta el nodo **Player** con un nodo **Country**.

Los nodos **League** representan las ligas en las que juegan los futbolistas, y cada una está asociada a un país específico a través de la relación **FROM_COUNTRY**, que conecta el nodo **League** con el correspondiente nodo **Country**.

Además, se han incorporado nodos **Season**, **Statistics** y **Transfer** para representar las temporadas, las estadísticas y las transferencias de los jugadores, respectivamente.

El nodo **Season** está vinculado a las estadísticas a través de la relación **IN_**, que permite identificar la temporada correspondiente a cada conjunto de estadísticas. Las estadísticas de un jugador específico se conectan a través de la relación **WITH**, y

¹<https://api-sports.io/>

²Específicamente, **BSON** en MongoDB, que es una extensión binaria de **JSON**

³https://gitlab.fing.edu.uy/bdnr2023_grupo_21/proyecto-final-bdnr-2023-grupo-21

el equipo en el que se generaron dichas estadísticas se identifica mediante la relación **STATS_FOR**. El nodo **Statistics** incluye información sobre el desempeño del jugador, como goles, asistencias, duelos ganados, entre otros.

El nodo **Transfer** está relacionado con el nodo **Season** a través de la relación **DURING**, que indica la temporada en la que se produjo la transferencia. La relación **FROM** conecta el nodo **Transfer** con el nodo **Player**, indicando el jugador que fue transferido. Las relaciones **IN** y **OUT** vinculan el nodo **Transfer** con los nodos **Team**, denotando el equipo de origen y de destino en la transferencia, respectivamente.

La Figura 1 proporciona una representación visual del modelo de grafo propuesto descripto.




Figura 1: Modelo de la base de datos de grafo.

IV. CONSULTAS

A continuación mostramos el resultado de diversas consultas realizadas para analizar el modelo de grafos resultante, donde se analiza con diversos recorridos posibles consultas de interés, enfocándonos en jugadores Uruguayos, equipos Uruguayos y también en transferencias a equipos Europeos.

IV-A. *Transferencia de jugadores nacidos en Uruguay a equipos Europeos*

Se obtienen a partir de las transferencias en las que un jugador nacido en Uruguay en las que va a jugar a un equipo de ligas Europeas. Se lista equipos Europeos, con la cantidad de transferencias de Uruguayos que realizaron en el período analizado, y una lista de sus nombres.

IV-A1. *Repetiendo jugadores:* Si un jugador se ha transferido al mismo equipo más de una vez, se lista tantas veces como lo haya hecho, retorna la cantidad de veces que el Equipo Europeo aceptó transferir a un Uruguayo en los últimos años.

```
MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)<-[:FROM]-(transfer:Transfer)
-[:IN]->(team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"]
WITH team.name AS Team, COUNT(transfer) AS NumTransfers, COLLECT(player.name) AS PlayerNames
RETURN Team, NumTransfers, PlayerNames
ORDER BY NumTransfers DESC
```

Team	NumTransfers	PlayerNames
Liverpool	7	Pablo Andrés Pereira Errandonea, Christian Leonel Rodríguez Fleitas, Maximiliano Felipe Montero Rodríguez, Cristian Martín Rodríguez Telis, Álvaro Maximiliano Arias Invernizzi
Deportivo La Coruna	4	Juan Ignacio González Brazeiro, Federico Santiago Valverde Dipetta, Diego Ifrán Sala, Fabián Larry Estoyanoff Poggio
Girona	4	Maximiliano Agustín Villa Pereira
Mallorca	3	Giovanni Alessandro González Apud, Pablo Domingo Cáceres Rodríguez
Espanyol	3	Juan Ángel Albín Leites, Iván Daniel Alonso Vallejo
Real Madrid	3	Federico Santiago Valverde Dipetta, Guillermo Varela Olivera
Real Sociedad	3	Diego Ifrán Sala
Manchester United	3	Guillermo Varela Olivera
Monaco	2	Sebastián César Helios Ribas Barbato, Ernesto Javier Chevantón Espinosa
Sporting Gijon	2	Jorge Giovani Zarfino Calandria, Guillermo Rafael De Amores Ravelo
Alaves	2	Carlos Nahuel Benavídez Protesoni
Atletico Madrid	2	Nicolás Javier Schiappacasse Oliva, Lucas Sebastián Torreira di Pascua
Valencia	2	Fabián Larry Estoyanoff Poggio, Maximiliano Gómez González
Arsenal	2	Lucas Sebastián Torreira di Pascua
West Ham	2	Diego Poyet González
Getafe	1	Juan Ángel Albín Leites
Strasbourg	1	Sebastián César Helios Ribas Barbato
Racing Santander	1	Adrián Martín Balboa Camacho
Watford	1	Miguel Ángel Britos Cabrera
Almeria	1	Ernesto Goñi Ameijenda
Sevilla	1	Ernesto Javier Chevantón Espinosa
QPR	1	Ernesto Javier Chevantón Espinosa
Paris Saint Germain	1	Diego Alfredo Lugano Morena
Malaga	1	Diego Alfredo Lugano Morena
West Brom	1	Diego Alfredo Lugano Morena
Valladolid	1	Fabián Larry Estoyanoff Poggio
Levante	1	Marcelo Josemir Saracchi Pintos
Celta Vigo	1	Maximiliano Gómez González
Las Palmas	1	Andrés Lamas Berdejillo
Toulouse	1	Mauro Daniel Goicoechea Furia

IV-A2. *Sin repetir jugadores:* Si un jugador se ha transferido al mismo equipo más de una vez, aparece listado solo una vez, muestra la cantidad de Uruguayos aceptados por la Liga Europa en los últimos años.

```
MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)<-[:FROM]-(transfer:Transfer)
-[:IN]->(team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"]
WITH team.name AS Team, COUNT(DISTINCT player.name) AS NumTransfers, COLLECT(DISTINCT player.name) AS PlayerNames
RETURN Team, NumTransfers, PlayerNames
ORDER BY NumTransfers DESC
```

Team	NumTransfers	PlayerNames
Liverpool	5	Pablo Andrés Pereira Errandonea, Christian Leonel Rodríguez Fleitas, Maximiliano Felipe Montero Rodríguez, Cristian Martín Rodríguez Telis, Álvaro Maximiliano Arias Invernizzi
Deportivo La Coruna	4	Juan Ignacio González Brazeiro, Federico Santiago Valverde Dipetta, Diego Ifrán Sala, Fabián Larry Estoyanoff Poggio
Mallorca	2	Giovanni Alessandro González Apud, Pablo Domingo Cáceres Rodríguez
Espanyol	2	Juan Ángel Albín Leites, Iván Daniel Alonso Vallejo
Monaco	2	Sebastián César Helios Ribas Barbato, Ernesto Javier Chevantón Espinosa
Sporting Gijon	2	Jorge Giovani Zarfino Calandria, Guillermo Rafael De Amores Ravelo
Real Madrid	2	Federico Santiago Valverde Dipetta, Guillermo Varela Olivera
Alaves	2	Carlos Nahuel Benavídez Protesoni, Iván Daniel Alonso Vallejo
Atletico Madrid	2	Nicolás Javier Schiappacasse Oliva, Lucas Sebastián Torreira di Pascua
Valencia	2	Fabián Larry Estoyanoff Poggio, Maximiliano Gómez González
Getafe	1	Juan Ángel Albín Leites
Strasbourg	1	Sebastián César Helios Ribas Barbato
Racing Santander	1	Adrián Martín Balboa Camacho
Watford	1	Miguel Ángel Britos Cabrera
Almeria	1	Ernesto Goñi Ameijenda
Sevilla	1	Ernesto Javier Chevantón Espinosa
QPR	1	Ernesto Javier Chevantón Espinosa
Real Sociedad	1	Diego Ifrán Sala
Manchester United	1	Guillermo Varela Olivera
Paris Saint Germain	1	Diego Alfredo Lugano Morena
Malaga	1	Diego Alfredo Lugano Morena
West Brom	1	Diego Alfredo Lugano Morena
Valladolid	1	Fabián Larry Estoyanoff Poggio
Arsenal	1	Lucas Sebastián Torreira di Pascua
Levante	1	Marcelo Josemir Saracchi Pintos
Girona	1	Maximiliano Agustín Villa Pereira
Celta Vigo	1	Maximiliano Gómez González
Las Palmas	1	Andrés Lamas Bervejillo
Toulouse	1	Mauro Daniel Goicoechea Furia
West Ham	1	Diego Poyet González

IV-B. Equipos Uruguayos que transfieren jugadores a equipos Europeos.

Vemos los equipos de ligas uruguayas que hayan transferido directamente jugadores a equipos de ligas europeas. Con esto podemos ver qué equipos uruguayos generan en los últimos años jugadores de interés para ser llamados por equipos Europeos.

```

MATCH (uruguayTeam:Team)-[:BELONGS_TO]->(uruguayanLeagues:League)
WHERE uruguayanLeagues.name IN ["Primera Division - Apertura", "Primera Division - Clausura", "Segunda Division"]
MATCH (uruguayTeam)-[:OUT]-(transfer:Transfer)-[:IN]->(europeanTeam:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"]
MATCH (transfer)-[:FROM]->(player:Player)
WITH uruguayTeam, europeanTeam, COLLECT(DISTINCT player.name) AS playerNames, SIZE(COLLECT(DISTINCT player.name)) AS
    transferCount
WITH uruguayTeam, COLLECT({europeanTeam: europeanTeam.name, transferCount: transferCount, playerNames: playerNames}) AS
    transfers, SUM(transferCount) AS totalTransfers
RETURN uruguayTeam.name AS uruguayTeam, transfers, totalTransfers
ORDER BY totalTransfers DESC

```

Equipo	Numero de Transferencias	Transferencias - Equipo Europeo
Penarol	4	Federico Santiago Valverde Dipetta - Real Madrid Giovanni Alessandro González Apud - Mallorca Fabián Larry Estoyanoff Poggio - Valencia Guillermo Varela Olivera - Manchester United
Club Nacional	3	Maximiliano Agustín Villa Pereira - Girona Juan Ángel Albín Leites - Getafe Juan Ángel Albín Leites - Espanyol
Tacuarembó	2	Christian Leonel Rodríguez Fleitas - Liverpool Cristian Martín Rodríguez Telis - Liverpool
CA River Plate	2	Nicolás Javier Schiappacasse Oliva - Atletico Madrid Iván Daniel Alonso Vallejo - Alaves
Danubio	1	Diego Ifrán Sala - Real Sociedad
Defensor Sporting	1	Maximiliano Gómez González - Celta Vigo
Central Espanol	1	Pablo Andrés Pereira Errandonea - Liverpool
Plaza Colonia	1	Juan Ignacio González Brazeiro - Deportivo La Coruna
Rampla Juniors	1	Pablo Andrés Pereira Errandonea - Liverpool
SUD America	1	Álvaro Maximiliano Arias Invernizzi - Liverpool

IV-C. Ciudades que generaron jugadores en las ligas Europeas

Podemos analizar las ciudades de origen uruguayas de los jugadores de liga Europea de la siguiente forma

```
MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)<-[:FROM]-(transfer:Transfer)-[:IN]->(team:Team)
-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"]
WITH player.birth_place AS BirthPlace, COLLECT(DISTINCT player.name) AS PlayerNames, COUNT(DISTINCT player) AS
PlayerCount
RETURN BirthPlace, PlayerCount, PlayerNames
ORDER BY PlayerCount DESC
```

BirthPlace	PlayerCount	PlayerNames
Montevideo	19	Pablo Andrés Pereira Errandonea,Giovanni Alessandro González Apud,Sebastián César Helios Ribas Barbato,Adrián Martín Balboa Camacho,Jorge Giovanni Zarfino Calandria,Miguel Ángel Britos Cabrera,Maximiliano Felipe Montero Rodríguez,Federico Santiago Valverde Dipetta,Ernesto Goñi Ameijenda,Pablo Domingo Cáceres Rodríguez,Carlos Nahuel Benavídez Protesoni,Nicolás Javier Schiappacasse Oliva,Iván Daniel Alonso Vallejo,Guillermo Varela Olivera,Fabián Larry Estoyanoff Poggio,Maximiliano Agustín Villa Pereira,Álvaro Maximiliano Arias Invernizzi,Andrés Lamas Bervejillo,Mauro Daniel Goicoechea Furia
Paysandú	3	Juan Ignacio González Brazeiro,Marcelo Josemir Saracchi Pintos,Maximiliano Gómez González
Salto	1	Juan Ángel Albín Leites
Tacuarembó	1	Christian Leonel Rodríguez Fleitas
San Jacinto	1	Guillermo Rafael De Amores Ravelo
Juan Lacaze	1	Ernesto Javier Chevantón Espinosa
Cerro Chato	1	Diego Ifrán Sala
Colonia Palma	1	Cristian Martín Rodríguez Telis
Canelones	1	Diego Alfredo Lugano Morena
Fray Bentos	1	Lucas Sebastián Torreira di Pascua
Zaragoza	1	Diego Poyet González

IV-D. Cantidad de jugadores uruguayos que jugaron en una liga Europea en cada Temporada

Podemos ver cuantos jugadores en los años que tenemos datos han sido de nacionalidad uruguay de los recolectados en los datos

```

MATCH (player:Player)-[:BORN_IN]->(country:Country {name: "Uruguay"})
MATCH (player)-[:WITH]->(statistics:Statistics)-[:STATS_FOR]->(team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"]
WITH team, league, player, statistics
MATCH (statistics)-[:IN]->(season:Season)
RETURN season.name AS SeasonName, COUNT(DISTINCT player) AS PlayerCount, COLLECT(DISTINCT player.name) AS PlayerNames
ORDER BY SeasonName;

```

SeasonName	PlayerCount	PlayerNames
2010-2011	2	Carlos Andrés Diogo Enseñat, Gustavo Adolfo Munúa Vera
2011-2012	8	Juan Ángel Albín Leites, Walter Gerardo Pandiani Urquiza, Pablo Domingo Cáceres Rodríguez, Sebastián Eguren Ledesma, Gustavo Adolfo Munúa Vera, Diego Ifrán Sala, Pablo César Pintos Cabral, Diego Alfredo Lugano Morena
2012-2013	2	Gustavo Adolfo Munúa Vera, Diego Alfredo Lugano Morena
2013-2014	1	Diego Ifrán Sala
2014-2015	2	Cristian Gabriel Rodríguez Barrotti, Diego Poyet González
2015-2016	3	Miguel Ángel Britos Cabrera, Álvaro Daniel Pereira Barragán, Diego Poyet González
2016-2017	2	Miguel Ángel Britos Cabrera, Facundo Jeremías Guichón Sisto
2017-2018	3	Miguel Ángel Britos Cabrera, Ramiro Guerra Pereyra, Guillermo Varela Olivera
2018-2019	2	Miguel Ángel Britos Cabrera, Mathías Olivera Miramontes
2019-2020	3	Agustín Maximiliano Rogel Paita, Ramiro Guerra Pereyra, Mathías Olivera Miramontes
2020-2021	7	Ronald Federico Araujo da Silva, Juan Ignacio González Brazeiro, Federico Santiago Valverde Dipetta, Lautaro de León Billar, Mathías Olivera Miramontes, Lucas Sebastián Torreira di Pascua, Maximiliano Gómez González
2021-2022	2	Martín Adrián Satriano Costa, Mathías Olivera Miramontes
2022-2023	5	Álvaro Daniel Rodríguez Muñoz, Ronald Federico Araujo da Silva, Federico Santiago Valverde Dipetta, Maximiliano Gómez González, Mathías Sebastián Suárez Suárez

IV-E. Edad de cada jugador en el momento de primera transferencia a Europa

Para cada jugador transferido de liga Uruguaya a Liga Europea, encontramos la edad que tenían al realizarse la primera transferencia a Europa, restando la fecha de la transferencia con la fecha de su nacimiento.

```

MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)<-[:FROM]-(transfer:Transfer)-[:OUT]->(
    uruguayanTeam:Team)-[:BELONGS_TO]->(uruguayanLeagues:League)
WHERE uruguayanLeagues.name IN ["Primera Division - Apertura", "Primera Division - Clausura", "Segunda Division"]
MATCH (transfer)-[:IN]->(europeanTeam:Team)-[:BELONGS_TO]->(europeanLeagues:League)
WHERE europeanLeagues.name IN ["Ligue 1", "La Liga", "Premier League"]
WITH player, min(date(transfer.date)) AS minTransferDate, date(player.birth_date).year AS birthYear, uruguayanTeam,
    europeanTeam
WITH player, minTransferDate.year - birthYear AS ageDifference, uruguayanTeam, europeanTeam, minTransferDate
ORDER BY ageDifference
RETURN player.name AS PlayerName, uruguayanTeam.name AS UruguayanTeam, europeanTeam.name AS EuropeanTeam, ageDifference
    AS AgeDifference, minTransferDate AS TransferDate

```

PlayerName	UruguayanTeam	EuropeanTeam	AgeDifference	TransferDate
Nicolás Javier Schiappacasse Oliva	CA River Plate	Atletico Madrid	17	2016-07-01
Federico Santiago Valverde Dipetta	Penarol	Real Madrid	18	2016-07-22
Juan Ángel Albín Leites	Club Nacional	Getafe	20	2006-07-01
Guillermo Varela Olivera	Penarol	Manchester United	20	2013-07-01
Maximiliano Agustín Villa Pereira	Club Nacional	Girona	20	2017-07-13
Iván Daniel Alonso Vallejo	CA River Plate	Alaves	21	2000-07-01
Maximiliano Gómez González	Defensor Sporting	Celta Vigo	21	2017-07-01
Christian Leonel Rodríguez Fleitas	Tacuarembó	Liverpool	22	2017-04-05
Diego Ifrán Sala	Danubio	Real Sociedad	23	2010-08-21
Pablo Andrés Pereira Errandonea	Central Espanol	Liverpool	26	2011-08-11
Fabián Larry Estoyanoff Poggio	Penarol	Valencia	26	2008-08-01
Juan Ángel Albín Leites	Club Nacional	Espanyol	27	2013-07-01
Juan Ignacio González Brasileiro	Plaza Colonia	Deportivo La Coruna	27	2020-09-14
Giovanni Alessandro González Apud	Penarol	Mallorca	28	2022-01-29
Álvaro Maximiliano Arias Invernizzi	SUD America	Liverpool	28	2016-07-04
Cristian Martín Rodríguez Telis	Tacuarembó	Liverpool	30	2015-07-20
Pablo Andrés Pereira Errandonea	Rampla Juniors	Liverpool	31	2016-01-04

IV-F. Jugadores no Uruguayos en equipos Uruguayos

Muestra las temporadas en las que hubieron jugadores en un equipo Uruguayo nacidos en un país extranjero, y cuantos jugadores de ese País pasaron por el equipo durante dichas temporadas.

```

MATCH (player:Player)-[:BORN_IN]->(birthCountry:Country)
WHERE birthCountry.name <> "Uruguay"
WITH player, birthCountry
MATCH (player)-[:WITH]->(statistic:Statistics)-[:STATS_FOR]->(team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Primera Division - Apertura", "Primera Division - Clausura", "Segunda Division"]
WITH player, birthCountry, team, statistic
MATCH (statistic)-[:IN]->(season:Season)
RETURN team.name AS Team, birthCountry.name AS BirthCountry, COUNT(DISTINCT player) AS PlayerCount, COLLECT(DISTINCT
season.name) AS Seasons
ORDER BY Team, PlayerCount DESC

```

Como el resultado de la consulta es muy grande, se decidió elegir los equipos uruguayos 'Club Nacional', 'Penarol' 'Danubio' y 'Boston River' para mostrar el resultado.

```

MATCH (player:Player)-[:BORN_IN]->(birthCountry:Country)
WHERE birthCountry.name <> "Uruguay"
WITH player, birthCountry
MATCH (player)-[:WITH]->(statistic:Statistics)-[:STATS_FOR]->(team:Team)
WHERE team.name IN ["Club Nacional", "Penarol", "Danubio", "Boston River"]
WITH player, birthCountry, team, statistic
MATCH (statistic)-[:IN]->(season:Season)
RETURN team.name AS Team, birthCountry.name AS BirthCountry, COUNT(DISTINCT player) AS PlayerCount, COLLECT(DISTINCT
season.name) AS Seasons
ORDER BY Team, PlayerCount DESC

```

Team	BirthCountry	PlayerCount	Seasons
Boston River	Argentina	3	2021-2022,2020-2021,2019-2020,2018-2019,2017-2018
Boston River	Brazil	3	2020-2021,2019-2020,2018-2019
Boston River	Venezuela	2	2020-2021,2018-2019
Boston River	Japan	1	2013-2014
Club Nacional	Argentina	7	2021-2022,2020-2021,2019-2020,2018-2019
Club Nacional	Colombia	1	2014-2015
Club Nacional	Peru	1	2013-2014
Danubio	Argentina	6	2021-2022,2022-2023,2020-2021,2019-2020,2017-2018,2012-2013
Danubio	Brazil	3	2017-2018,2015-2016,2013-2014
Danubio	Paraguay	1	2020-2021
Danubio	Australia	1	2013-2014
Danubio	Colombia	1	2023-2024
Penarol	Argentina	5	2022-2023,2019-2020,2018-2019,2017-2018,2016-2017,2015-2016,2013-2014
Penarol	Hungary	1	2020-2021
Penarol	Spain	1	2019-2020
Penarol	Colombia	1	2019-2020
Penarol	Brazil	1	2017-2018

IV-G. Vuelta de jugadores al equipo Uruguayo

Es listado el jugador que se va de un equipo de Uruguay a un equipo Europeo en una transferencia, y luego retorna al mismo equipo de Uruguay.

```

MATCH (player:Player) <-[:FROM]-(transfer:Transfer)-[:OUT]->(uruguayanTeam:Team)-[:BELONGS_TO]->(uruguayanLeagues:League)
WHERE uruguayanLeagues.name IN ["Primera Division - Apertura", "Primera Division - Clausura", "Segunda Division"]
MATCH (transfer)-[:IN]->(europeanTeam:Team)-[:BELONGS_TO]->(europeanLeagues:League)
WHERE europeanLeagues.name IN ["Ligue 1", "La Liga", "Premier League"]
MATCH (player) <-[:FROM]-(transfer2:Transfer)-[:IN]->(uruguayanTeam:Team)
WHERE date(transfer.date) < date(transfer2.date) AND transfer <> transfer2
MATCH (transfer2)-[:OUT]->(outTeam2:Team)
RETURN player.name AS PlayerName, transfer.date AS FirstTransferDate, uruguayanTeam.name AS UruguayTeamOut,
       europeanTeam.name AS EuropeanTeamIn, transfer2.date AS SecondTransferDate, outTeam2.name AS TeamOut, uruguayanTeam
       .name AS UruguayTeamIn

```

PlayerName	FirstTransferDate	UruguayTeamOut	EuropeanTeamIn	SecondTransferDate	TeamOut	UruguayTeamIn
Pablo Andrés Pereira Errandonea	2011-08-11	Central Espanol	Liverpool	2013-02-05	Correcaminos Uat	Central Espanol
Álvaro Maximiliano Arias Invernizzi	2016-07-04	SUD America	Liverpool	2018-02-12	River Plate	SUD America
Juan Ángel Albín Leites	2006-07-01	Club Nacional	Getafe	2013-01-17	Espanyol	Club Nacional
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2010-07-10	Panionios	Penarol
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2012-01-05	Panionios	Penarol
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2017-08-18	Fenix	Penarol
Guillermo Varela Olivera	2013-07-01	Penarol	Manchester United	2017-08-14	Manchester United	Penarol
Álvaro Maximiliano Arias Invernizzi	2016-07-04	SUD America	Liverpool	2018-02-12	River Plate	SUD America
Juan Ángel Albín Leites	2006-07-01	Club Nacional	Getafe	2013-01-17	Espanyol	Club Nacional
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2010-07-10	Panionios	Penarol
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2012-01-05	Panionios	Penarol
Fabián Larry Estoyanoff Poggio	2008-08-01	Penarol	Valencia	2017-08-18	Fenix	Penarol
Guillermo Varela Olivera	2013-07-01	Penarol	Manchester United	2017-08-14	Manchester United	Penarol
Álvaro Maximiliano Arias Invernizzi	2016-07-04	SUD America	Liverpool	2018-02-12	River Plate	SUD America

IV-H. Top diez jugadores Uruguayos en Europa mejor posicionados según API-Sports

API-Sports calcula para los jugadores de mayor interés una estadística 'games_rating', donde se pone puntaje según su rendimiento en varias estadísticas a lo largo de una temporada. De jugadores nacidos en Uruguay que han jugado en equipo de Liga Europea tenemos dicha estadística, por lo que podemos ordenarlos según el mejor puntaje que tengan entre todas sus temporadas.

```

MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)-[:WITH]->(statistic:Statistics)-[:STATS_FOR]->(
    team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"] AND statistic.games_rating IS NOT NULL
MATCH (statistic)-[:IN]->(season:Season)
WITH player, statistic, team, season
ORDER BY statistic.games_rating DESC
RETURN player.name AS PlayerName, COLLECT({season: season.name, team: team.name, games_rating: statistic.games_rating})
[0] AS BestStatistic
ORDER BY BestStatistic DESC
LIMIT 10

```

Jugador	Games_rating	Temporada	Equipo
Federico Santiago Valverde Dipetta	7.273529	2022-2023	Real Madrid
Ronald Federico Araujo da Silva	7.063636	2022-2023	Barcelona
Maximiliano Gómez González	6.980645	2020-2021	Valencia
Mathías Olivera Miramontes	6.962500	2021-2022	Getafe
Miguel Ángel Britos Cabrera	6.914814	2016-2017	Watford
Álvaro Daniel Rodríguez Muñoz	6.740000	2022-2023	Real Madrid
Lucas Sebastián Torreira di Pascua	6.694736	2020-2021	Atletico Madrid
Agustín Maximiliano Rogel Paita	6.533333	2019-2020	Toulouse
Álvaro Daniel Pereira Barragán	6.350000	2015-2016	Getafe
Lautaro de León Billar	6.300000	2020-2021	Celta Vigo

IV-I. Cantidad de goles de delanteros Uruguayos

Para cada delantero Uruguayo con estadística obtenida en la Liga Europea en el período dado, obtiene la cantidad de goles que hizo en esa temporada.

```

MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)-[:WITH]->(statistic:Statistics)-[:STATS_FOR]->(
  team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"] AND statistic.games_rating IS NOT NULL
MATCH (statistic)-[:IN]->(season:Season)
WITH player, statistic, team, season
ORDER BY statistic.games_rating DESC
RETURN player.name AS PlayerName, COLLECT({season: season.name, team: team.name, games_rating: statistic.games_rating})
[0] AS BestStatistic
ORDER BY BestStatistic DESC

```

PlayerName	Season	TeamName	GoalsTotal
Maximiliano Gómez González	2020-2021	Valencia	7
Martín Adrián Satriano Costa	2021-2022	Stade Brestois 29	4
Walter Gerardo Pandiani Urquiza	2011-2012	Espanyol	3
Álvaro Daniel Rodríguez Muñoz	2022-2023	Real Madrid	1
Diego Ifrán Sala	2011-2012	Real Sociedad	1
Diego Ifrán Sala	2013-2014	Real Sociedad	0
Lautaro de León Billar	2020-2021	Celta Vigo	0
Maximiliano Gómez González	2022-2023	Valencia	0

IV-J. Estadísticas de defensores Uruguayos

Nos permite visualizar las estadísticas de defensores nacidos en Uruguay con campos no vacíos en cada temporada que han jugado en ligas Europeas, ya que no hay estadísticas con estos campos llenos para defensores en las Ligas Uruguayas en API-Sports.

```

MATCH (country:Country {name: "Uruguay"})<-[:BORN_IN]-(player:Player)-[:WITH]->(statistic:Statistics)-[:STATS_FOR]->(
  team:Team)-[:BELONGS_TO]->(league:League)
WHERE league.name IN ["Ligue 1", "La Liga", "Premier League"] AND statistic.games_position = "Defender" AND statistic.
  duels_total IS NOT NULL AND statistic.duels_won IS NOT NULL
MATCH (statistic)-[:IN]->(season:Season)
RETURN DISTINCT player.name AS PlayerName, season.name AS Season, team.name AS TeamName,
  statistic.duels_total AS DuelsTotal, statistic.duels_won AS DuelsWon,
  statistic.cards_yellow AS CardsYellow, statistic.cards_red AS CardsRed,
  statistic.passes_total AS PassesTotal, statistic.passes_key AS PassesKey,
  statistic.passes_accuracy AS PassesAccuracy
ORDER BY DuelsTotal desc

```

PlayerName	Season	TeamName	DuelsTotal	DuelsWon	CardsYellow	CardsRed	PassesTotal	PassesAccuracy
Mathías Olivera Miramontes	2021-2022	Getafe	521	263	10	0	1052	23
Mathías Olivera Miramontes	2020-2021	Getafe	340	188	6	0	831	18
Miguel Ángel Britos Cabre-ra	2016-2017	Watford	249	143	8	0	704	65
Miguel Ángel Britos Cabre-ra	2015-2016	Watford	177	109	7	0	534	62
Mathías Olivera Miramontes	2018-2019	Getafe	177	92	5	0	178	49
Mathías Olivera Miramontes	2019-2020	Getafe	161	94	6	0	323	52
Ronald Federico Araujo da Silva	2022-2023	Barcelona	151	97	6	1	1253	50
Ronald Federico Araujo da Silva	2020-2021	Barcelona	135	99	3	0	1065	40
Miguel Ángel Britos Cabre-ra	2017-2018	Watford	85	50	3	1	313	73
Álvaro Daniel Pereira Barragán	2015-2016	Getafe	49	23	3	0	82	51
Miguel Ángel Britos Cabre-ra	2018-2019	Watford	20	11	0	0	47	70

IV-K. Dado un jugador y una fecha de transferencia, obtenemos sus compañeros de equipo

Sin perder generabilidad, se elige al jugador Diego Andrés Martiñones Rus y la fecha de transferencia 2020-08-03 para realizar la consulta y obtener sus compañeros del equipo del que se retira y al que entra.

IV-K1. Obtenemos sus compañeros del equipo que se retira:

```

MATCH (player:Player {name: "Diego Andres Martiones Rus"})
MATCH (transfer:Transfer)-[:FROM]->(player)
MATCH (transfer)-[:OUT]->(teamViejo:Team)
MATCH (player)-[:WITH]->(:Statistics)-[:IN]->(seasonTransfer:Season)
MATCH (player)-[:WITH]->(:Statistics)-[:STATS_FOR]->(teamViejo)
MATCH (playerCompanion)-[:WITH]->(:Statistics)-[:IN]->(seasonTransfer)
MATCH (playerCompanion)-[:WITH]->(:Statistics)-[:STATS_FOR]->(teamViejo)
MATCH (transfer)-[:DURING]->(seasonTransfer)
WHERE player <> playerCompanion AND transfer.date = "2020-08-03"
RETURN DISTINCT playerCompanion.name AS CompanionName, teamViejo.name as TeamViejo, seasonTransfer.name as Season,
transfer.date as transferDate

```

CompanionName	TeamViejo	Season	transferDate
Leandro Joaquín Rodríguez Telechea	Danubio	2020-2021	2020-08-03
Jorge Daniel Graví Piñeiro	Danubio	2020-2021	2020-08-03
Fredy Joel Martínez Mancilla	Danubio	2020-2021	2020-08-03
Ignacio María González Gatti	Danubio	2020-2021	2020-08-03
Esteban Néstor Conde Quintana	Danubio	2020-2021	2020-08-03
Matías Lionel Fritzler	Danubio	2020-2021	2020-08-03
Matías Nicolás Deorta Azar	Danubio	2020-2021	2020-08-03
Julio César Domínguez Castillo	Danubio	2020-2021	2020-08-03
Leandro Onetto Baccino	Danubio	2020-2021	2020-08-03
Diego Ifrán Sala	Danubio	2020-2021	2020-08-03
Mauricio Bernardo Victorino Dansilio	Danubio	2020-2021	2020-08-03
Carlos Javier Grossmüller	Danubio	2020-2021	2020-08-03
Federico Alfredo Cristóforo Pepe	Danubio	2020-2021	2020-08-03
Renzo Ramírez Maidana	Danubio	2020-2021	2020-08-03
Ernesto Goñi Ameijenda	Danubio	2020-2021	2020-08-03
Juan Ignacio González Brazeiro	Danubio	2020-2021	2020-08-03
Pablo Martín Silva Silva	Danubio	2020-2021	2020-08-03
Juan Manuel Olivera López	Danubio	2020-2021	2020-08-03
Sergio Nicolás Leites Montes	Danubio	2020-2021	2020-08-03
Damián Alejandro Malrechauffe Verdún	Danubio	2020-2021	2020-08-03
Carlos Alberto Canobbio Bentaberry	Danubio	2020-2021	2020-08-03
Líber Daniel Quiñones Prieto	Danubio	2020-2021	2020-08-03
Gonzalo Fabián Porras Burghi	Danubio	2020-2021	2020-08-03
Jorge Carlos Zambrana Echagüe	Danubio	2020-2021	2020-08-03

IV-K2. Obtenemos sus compañeros del equipo que al que entra:

```

MATCH (player:Player {name: "Diego Andres Martiones Rus"})
MATCH (transfer:Transfer)-[:FROM]->(player)
MATCH (transfer)-[:IN]->(teamNuevo:Team)
MATCH (player)-[:WITH]->(:Statistics)-[:IN]->(seasonTransfer:Season)
MATCH (player)-[:WITH]->(:Statistics)-[:STATS_FOR]->(teamNuevo)
MATCH (playerCompanion)-[:WITH]->(:Statistics)-[:IN]->(seasonTransfer)
MATCH (playerCompanion)-[:WITH]->(:Statistics)-[:STATS_FOR]->(teamNuevo)
MATCH (transfer)-[:DURING]->(seasonTransfer)
WHERE player <> playerCompanion AND transfer.date = "2020-08-03"
RETURN DISTINCT playerCompanion.name AS CompanionName, teamNuevo.name as teamNuevo, seasonTransfer.name as Season,
transfer.date as transferDate

```

CompanionName	teamNuevo	Season	transferDate
Richard Eduardo Rodríguez Rangel	Rampla Juniors	2020-2021	2020-08-03
Jonathan Sebastián Charquero López	Rampla Juniors	2020-2021	2020-08-03
Obasiarinse Idehaloise Uade	Rampla Juniors	2020-2021	2020-08-03
Adrián Ezequiel Dorta Hernández	Rampla Juniors	2020-2021	2020-08-03
Cristian Emanuel Ferreira Molina	Rampla Juniors	2020-2021	2020-08-03
José Augusto Sardón	Rampla Juniors	2020-2021	2020-08-03
Santiago Ignacio Vega Maidana	Rampla Juniors	2020-2021	2020-08-03
Facundo Elías Mesones Gamarra	Rampla Juniors	2020-2021	2020-08-03
Franco Guillermo Gentile Rotondaro	Rampla Juniors	2020-2021	2020-08-03
Óscar Padula Castro Rodríguez	Rampla Juniors	2020-2021	2020-08-03
Eduardo Alexander Chocho Silva	Rampla Juniors	2020-2021	2020-08-03
Santiago Fernández Idiart	Rampla Juniors	2020-2021	2020-08-03
Bryan Federico Bautista López	Rampla Juniors	2020-2021	2020-08-03
Luis Iván Cartés Maidana	Rampla Juniors	2020-2021	2020-08-03
Mateo Abeijón Porto	Rampla Juniors	2020-2021	2020-08-03
Juan Ángel Albín Leites	Rampla Juniors	2020-2021	2020-08-03
Rodrigo Ezequiel Vidal Piñeyro	Rampla Juniors	2020-2021	2020-08-03
Jonathan Alexander Ramírez Silva	Rampla Juniors	2020-2021	2020-08-03
Jorge Luis Trinidad Ferreira	Rampla Juniors	2020-2021	2020-08-03
Carlos Rodrigo Núñez Techera	Rampla Juniors	2020-2021	2020-08-03
Charles Mijael Zoryez Arbiza	Rampla Juniors	2020-2021	2020-08-03
Rodrigo Alejandro Ferreyra Ocampo	Rampla Juniors	2020-2021	2020-08-03
Bernardo Enzo Long Baccino	Rampla Juniors	2020-2021	2020-08-03
Elías Ezequiel Umeres	Rampla Juniors	2020-2021	2020-08-03
Luis Santiago Gáspari Vázquez	Rampla Juniors	2020-2021	2020-08-03
Matías Wolf Rigoletto Arévalo	Rampla Juniors	2020-2021	2020-08-03
Álex Maximiliano Silva Garrel	Rampla Juniors	2020-2021	2020-08-03

IV-L. Dado un jugador y una fecha de transferencia, obtenemos estadística

Podemos pedir datos de como le fue antes y despues de cambiarse de equipo, por ejemplo, pidiendo la cantidad de goles que hizo en la misma temporada antes y despues de transferirse. Sin perder generabilidad, se elige al jugador Fabián Larry Estoyanoff Poggio y la fecha de transferencia 2017-08-18 para realizar la consulta.

```

MATCH (player:Player {name: "Fabian Larry Estoyanoff Poggio"})
MATCH (transfer:Transfer)-[:FROM]->(player)
MATCH (transfer)-[:IN]->(teamNuevo:Team)
MATCH (transfer)-[:OUT]->(teamViejo:Team)
MATCH (transfer)-[:DURING]->(seasonTransfer:Season)
MATCH (estadisticaEquipoViejo:Statistics)<-[:WITH]- (player)
MATCH (estadisticaEquipoViejo)-[:IN]->(seasonTransfer)
MATCH (estadisticaEquipoViejo)-[:STATS_FOR]->(teamViejo)
MATCH (estadisticaEquipoNuevo:Statistics)<-[:WITH]- (player)
MATCH (estadisticaEquipoNuevo)-[:IN]->(seasonTransfer)
MATCH (estadisticaEquipoNuevo)-[:STATS_FOR]->(teamNuevo)
WHERE transfer.date = "2017-08-18"
RETURN player.name AS PlayerName, teamViejo.name AS TeamViejoName, estadisticaEquipoViejo.goals_total AS
GoalsTotalViejo,
teamNuevo.name AS TeamNuevoName, estadisticaEquipoNuevo.goals_total AS GoalsTotalNuevo

```

PlayerName	TeamViejoName	GoalsTotalViejo	TeamNuevoName	GoalsTotalNuevo
Fabián Larry Estoyanoff Poggio	Fenix	13	Penarol	3

V. CONCLUSIONES Y TRABAJO FUTURO

El objetivo principal de este proyecto ha sido introducir la herramienta Neo4j, modelando la realidad de jugadores Uruguayos en el mundo a través del uso de grafos. Tras almacenar esta información en la base de datos de Neo4j, se comenzó extrayendo datos relevantes e interesantes, explorando cada nodo y sus relaciones. Se considera que el objetivo general de aprendizaje se ha cumplido.

El proyecto tenía objetivos específicos, como responder a preguntas detalladas acerca de las actividades de los jugadores en relación con las transferencias a tres de las grandes ligas ("Premier League", "Ligue 1", "La Liga") en el periodo de 2010 al

presente. Se logró determinar quiénes eran estos jugadores, a qué ligas se trasladaron, sus edades, desde qué equipo uruguayo se marcharon y de qué departamento o ciudad procedían. También se realizó otros tipos de búsquedas, como el jugador más joven que se transfirió, qué equipo uruguayo tiene más transferencias, y si hay alguna liga o equipo que muestra una preferencia particular por los jugadores de aquí.

Se intentó incorporar las estadísticas de los jugadores para poder evaluar su rendimiento tanto local como internacional, así como los jugadores que los acompañaban. A pesar de que la API proporcionaba abundante información sobre las estadísticas actuales de equipos y jugadores, no conseguimos obtener datos equivalentes para años anteriores. Realizamos una extensa búsqueda para encontrar otra fuente para estos datos y luego intentamos integrarlos con los que ya teníamos, pero sin éxito. Por lo tanto, tuvimos que trabajar con los datos que sí conseguimos, lo que resultó en la obtención de resultados para algunos jugadores y no para otros, por lo que nos enfocamos en aquellos para los que sí teníamos información.

Uno de los desafíos que fueron enfrentados fue la carga de datos. Se tenía dos métodos para realizarlo: a través de archivos CSV o insertando manualmente cada consulta. Después de investigar, se encontró que la carga nodo por nodo era ligeramente más rápida que la realizada a través de CSV, por lo que se reorganizó los datos que teníamos en MongoDB, separados cada colección y dentro de ellas se obtuvo cada nodo. Luego se tuvo que pensar cómo realizaríamos las asociaciones entre ellos. Por ejemplo, las estadísticas que se obtuvo fueron a través de un `script.py` que las solicitaba por temporada para cada equipo y de cada equipo para cada jugador. Esto resultó en un enorme JSON, que fuimos desglosando mediante consultas, tratando de mantener las referencias de cada uno y no perder de quién era cada estadística, para quién y en qué temporada.

Se realizó otro programa para conectar con la base de datos de Neo4j y extraer los datos de MongoDB. Se hizo con la mayor cautela posible, insertando nodos de prueba, verificando su corrección y luego eliminándolos, hasta alcanzar un cierto nivel de robustez y confiabilidad para que al dejar el `script` corriendo por sí solo, insertara la información correctamente. La carga que más tiempo llevó fue de 9 horas, al cargar las estadísticas con sus relaciones.

Sin embargo, a pesar de la cautela, se encontró un gran problema de duplicación de nodos y relaciones entre nodos dobles e incluso triples. Esto se debió a la información proporcionada por la API, la cual era inviable de revisar nodo por nodo, pero generó estadísticas de los mismos jugadores más de una vez y para diferentes equipos. Este problema llevó a tener que rehacer parte del trabajo para encontrar una forma de eliminar los nodos duplicados sin tener que recargar todos los datos o eliminar todos los nodos duplicados, ya que esto habría requerido un tiempo adicional para cargarlos de nuevo.

En conclusión, se cree que los problemas encontrados no se deben directamente a la gestión de Neo4j o a la base de datos MongoDB, sino más bien a problemas relacionados con la obtención de los datos a trabajar, lo cual no es una tarea menor. En cuanto a los resultados que se obtuvieron al trabajar con cualquiera de las dos bases de datos, se considera que ambas son herramientas potentes y flexibles. Sin embargo, debido a su flexibilidad, es necesario tener mucho cuidado al usarlas, ya que los controles que se deben realizar deben ser lógicos antes de ingresar a la base de datos. Ambas bases de datos pueden almacenar documentos exactamente iguales pero con un `_id` interno diferente, y es muy fácil e intuitivo insertar datos en cualquiera de ellas, estén relacionados con otros nodos o documentos o no. En cuanto a su potencia, ambas bases de datos son rápidas y capaces de realizar consultas en segundos incluso con una gran carga de datos. En particular, la herramienta Neo4j AuraDB ofrece una interfaz intuitiva y amigable para intercambiar datos y detectar posibles errores.

Como trabajo futuro, puede ser beneficioso encontrar una fuente de datos más robusta para poder poblar la base con más estadísticas y más ligas, como la liga italiana y la alemana. Además, sería interesante poder ver las transferencias de jugadores uruguayos desde años anteriores a 2010, para poder incluir a jugadores como Luis Suárez, Edinson Cavani y muchos otros.